QMF Sydney 2006: Generating Quantile Functions using Computer Algebra: Distributional Alchemy & other tools for Copula & Monte Carlo Simulations

William T. Shaw, Financial Mathematics Group
King's College, London
william.shaw@kcl.ac.uk
http://www.mth.kcl.ac.uk/staff/w_shaw.html

Introduction

Given a distribution function $F_X(x)$, a simple means of simulation is to set

$$X = F_X^{-1}(U) \tag{1}$$

where U is a sample from the uniform distribution on [0, 1]. Making U is compsci probem. The inverse CDF: $F_X^{-1} = Q_{F_X}$ is the Quantile Function associated with the distribution. We do not have to do this, witness the use of Box-Muller, Polar-Marsaglia methods for the Normal case, and its extension to Student by Bailey. But it is very useful if we can, particularly if we are working with algorithms based on hypercube-filling quasi-Monte-Carlo methods, or in particular copula methods.

Quantile Functions and Copulas

For the case of copula-based simulation, we make a sample (X_1, X_2) from a given bivariate (in general multivariate) distribution, then form a sample from the associated copula:

$$\{U_1, U_2\} = \{F_{X_1}(X_1), F_{X_2}(X_2)\} \tag{2}$$

Then to get samples with marginals with CDFs G_i:

$$\{Y_1, Y_2\} = \{G_1^{-1}[U_1], G_2^{-1}[U_2]\} = \{Q_{G_1}[U_1], Q_{G_2}[U_2]\} \tag{3}$$

It is always helpful to know quantile functions. But note also we do not need the U_i, at least when we are dealing with samples from a real distribution, for it would suffice to understand the *composite* mappings in the following:

$$\{Y_1, Y_2\} = \{G_1^{-1}[F_{X_1}(X_1)], G_2^{-1}[F_{X_2}(X_2)]\} \tag{4}$$

This is one of several motivations for considering composite maps of the form $y = G^{-1}[F(x)]$, where F, G are CDFs.

Topics of the talk - see how far we get

Direct Construction of Quantile Functions themselves

- **Example 1: Recap Direct Student (Shaw, JCF, Vol 9, 4, pp.37-73,2006)**

Distributional Alchemy - from one Quantile Function to another

- **Example 1: Sample Transmutation: Student from Normal via**

  The "Exact Cornish-Fisher Map"

- **Example 2: Rank Transmutation: Introducing Skewness to Any Distribution without Gram-Charlier issues**

Construction of Quantile Functions themselves

Construction example 1:
Student Quantile Functions, iCDFs for even n

We can give special methods for the Student T distribution CDF when the degrees of freedom n is an even integer. Note EP's talk emphasized role of T_4 case. First we have an iterative specification of the CDF

```
a[0, n_] := Gamma[(n + 1) / 2] / Gamma[n / 2] / Sqrt[n Pi];
a[k_, n_] := a[k, n] = (n - 2 k) / n / (2 k + 1) a[k - 1, n];

TCDF[n_, x_] := 1 / 2 +
   x * Sum[a[p, n] x^(2 p), {p, 0, n / 2 - 1}] / (1 + x^2 / n)^(1 / 2 (n - 1));

{Simplify[TCDF[4, x]], Simplify[D[TCDF[4, x], x]], TCDF[4, -300] // N, TCDF[4, 300] // N}
```

$$\left\{\frac{x^3+6x+(x^2+4)^{3/2}}{2(x^2+4)^{3/2}},\ \frac{12}{(x^2+4)^{5/2}},\ 3.70343\times 10^{-10},\ 1.\right\}$$

Solve TCDF[n, x] $=$ u , a polynomial problem! Let $a = 4u(1-u)$ and $p = n + x^2$

```
P[n_, x_] := x * Sum[a[p, n] x^(2 p), {p, 0, n / 2 - 1}];
R[n_, p_] := Expand[PowerExpand[
    4 * n^(n - 1) (P[n, x])^2 + p^(n - 1) (a - 1) /. x → Sqrt[p - n]]] ;
TraditionalForm[Table[{2 k, R[2 k, p] == 0}, {k, 1, 7}]]
```

$$\begin{pmatrix}
2 & a p - 2 = 0 \\
4 & a p^3 - 12 p - 16 = 0 \\
6 & a p^5 - 135 p^2 - \frac{1215 p}{4} - \frac{2187}{2} = 0 \\
8 & a p^7 - 2240 p^3 - 7168 p^2 - 35840 p - 204800 = 0 \\
10 & a p^9 - \frac{196875 p^4}{4} - \frac{1640625 p^3}{8} - \frac{10546875 p^2}{8} - \frac{615234375 p}{64} - \frac{2392578125}{32} = 0 \\
12 & a p^{11} - 1347192 p^5 - 6928416 p^4 - 54561276 p^3 - 484989120 p^2 - 4583147184 p - 44998172352 = 0 \\
14 & a p^{13} - \frac{353299947 p^6}{8} - \frac{17311697403 p^5}{64} - \frac{40393960607 p^4}{16} - \frac{848273172747 p^3}{32} - \frac{18893357029365 p^2}{64} - \frac{872873094756663 p}{256} - \frac{5170094484327927}{128} = 0
\end{pmatrix}$$

and so on. We get interesting family of sparse polynomials. Almost half the coefficients are missing. Two exact solutions $n = 2, 4$, and $n = 6, 8, 10, \ldots$ easy by Newton-Raphson! (See JCF 2006, E. Platen's new book, for 1,2,4!)

Distributional Alchemy

Sample Transmutation

Given one "base" distribution function, say $\Phi(x)$ - the base might typically be Normal - and another distribution $F(x)$, we define a *sample transmutation mapping* T_S by the identity

$$F^{-1}[U] = T_S(\Phi^{-1}(U)),\ i.e.,\ \ T_S(z) = F^{-1}(\Phi(z)) = Q_F(\Phi(z)) \tag{5}$$

where $0 \le U \le 1$ and z is in its appropriate range (the real line in the Normal case). So if we have the Φ Quantile function we can get the Q_F quantile by post-applying T_S. This function "transmutes" samples from one distribution into samples from another. We might have a decent expression for Φ but Q_F may well be hard to nail down. But we will show some cunning computer algebra for getting series, and we can compose this with a series for Φ to get a good series for T_S.

This is not a new idea. It has previously found expression in the asymptotic setting via the use of Cornish-Fisher expansions, but we will now see how to use it in an essentially exact setting via the use of computer symbolic algebra.

Rank Transmutation

There is no particularly good reason why a transmutation mapping should be applied after applying a standard quantile rather than before. So we can define a corresponding *rank transmutation mapping* T_R by the following relationship

$$F^{-1}[U] = \Phi^{-1}(T_R(U)),\ \ i.e.,\ \ T_R(u) = \Phi(F^{-1}(z)) = \Phi(Q_F(u)) \tag{6}$$

This will allow us to introduce modulations into a distribution in an exact way, and potentially avoid the use of asymptotic (Edgeworth/Gram-Charlier [EGC]) methods and their problems. Note that equation (6) only makes sense if the two distributions have the same sample space.

Why think about these maps?

There are several reasons, some of which will represent a mathematical formalization of some things already being done by practitioners:

1. It helps us generate new (hard) Quantile functions from old (easy) ones for easy, QMC or copula applications;
2. We do not have to use the unit interval as a standard domain - we can change variables e.g. to Gaussian real line;
3. We can transmute a given sample to assess distributional risk in pricing/risk calculation, and avoid MC noise in much as the same way as Greeks are calculated in Monte Carlo;
4. Introduce skewness to standard distributions in a natural way;
5. Take the idea of QQ plot and transform it from a distributional assessment tool to a PDF generation tool;
6. Avoid pecularities of GC methods such as negative PDFs;s
7. Our original example based on copula simulation.

Example1: Sample Transmutation: Student from Normal via the "Exact Cornish-Fisher Map"

We consider the Student T distribution in the notation of Shaw (2006). Given $0 < u < 1$, we set

$$v = \frac{\left(u - \frac{1}{2}\right)\sqrt{n\pi}\;\Gamma\left(\frac{n}{2}\right)}{\Gamma\left(\frac{n+1}{2}\right)} \tag{7}$$

The quantile equation is then

$$v = \int_0^x \left(1 + \frac{s^2}{n}\right)^{-\frac{1}{2}(n+1)} ds = x\,{}_2F_1\left(\frac{1}{2}, \frac{n+1}{2}; \frac{3}{2}; -\frac{x^2}{n}\right) = F(x, n) \tag{8}$$

```
Off[General::spell1];
Off[Series::esss];
```

```
F[x_, n_] := x * Hypergeometric2F1[1 / 2, (n + 1) / 2, 3 / 2, - (x ^ 2 / n)];
SCDF[x_, n_] := 1 / 2 + 1 / Sqrt[n Pi] * Gamma[(n + 1) / 2] / Gamma[n / 2] F[x, n]
```

```
Simplify[D[SCDF[x, n], x]]
```

$$\frac{\left(\frac{x^2+n}{n}\right)^{-\frac{n}{2}-\frac{1}{2}} \Gamma\left(\frac{n+1}{2}\right)}{\sqrt{n} \sqrt{\pi} \Gamma\left(\frac{n}{2}\right)}$$

```
Integrate[%, {x, -Infinity, Infinity}, Assumptions → n > 0]
```

1

The inversion of a series for such a CDF can be carried out step by step in any computer algebra system, following the methods described in on-line information at KCL web site. In *Mathematica* you can sometimes just ask for the inverse as a series:

```
Map[Factor, InverseSeries[Series[F[x, n], {x, 0, 9}], v]]
```

$$v + \frac{(n+1)v^3}{6n} + \frac{(n+1)(7n+1)v^5}{120n^2} + \frac{(n+1)(127n^2+8n+1)v^7}{5040n^3} + \frac{(n+1)(4369n^3-537n^2+135n+1)v^9}{362880n^4} + O(v^{10})$$

With a bit more work a corresponding tail series can be developed. Requires a little intervention. (See Shaw, J. Comp Fin. 2006 for details). So we have the Quantile function for general real n. What we want to discuss here is the sample transmutation mapping.

One example of what sample transmutation then gives us is a relatively painless way of extracting a high order Cornish-Fisher expansion, and even a non-asymptotic form of it. Compare the following with the usual rash of high-order moments and Hermite functions in the case of the Student:

$$\Phi(z) - \frac{1}{2} = \frac{1}{2} \operatorname{erf}\left(\frac{z}{\sqrt{2}}\right) \tag{9}$$

```
Phi[z_] := 1/2(1+Erf[z/Sqrt[2]]);
PhiMinusHalf[z_] := Erf[z/Sqrt[2]]/2;
v[pmh_,n_]:=(pmh*Sqrt[n*Pi]*Gamma[n/2])/Gamma[(n+1)/2]
```

and we recall the form of $F(x, n)$ and set

```
QuantileF[v_, n_, truncation_] :=
  InverseSeries[Series[F[x, n], {x, 0, truncation}], v];
Transmutation[z_, n_, trunca_, truncb_] :=
 Module[{QF, toexp}, QF = QuantileF[v, n, trunca];
  toexp = Normal[QF /. {v → v[PhiMinusHalf[z], n], m → n}];
  Series[toexp, {z, 0, truncb}]]
```

```
rawform = Transmutation[z, n, 10, 10];
CornishFisherExpansion = Map[
  Together, Normal[Series[rawform, {n, Infinity, 4}]]]
```

$$z+\frac{z^3+z}{4n}+\frac{5z^5+16z^3+3z}{96n^2}+\frac{3z^7+19z^5+17z^3-15z}{384n^3}+\frac{79z^9+776z^7+1482z^5-1920z^3-945z}{92160n^4}$$

That can be found in Abramowitz and Stegun! But you wont find this one anywhere (prove me wrong!). Even my laptop has to think for a bit.

```
rawform = Transmutation[z, n, 20, 20];
CornishFisherExpansion = Map[
  Together, Normal[Series[rawform, {n, Infinity, 9}]]]
```

$$z+\frac{z^3+z}{4n}+\frac{5z^5+16z^3+3z}{96n^2}+\frac{3z^7+19z^5+17z^3-15z}{384n^3}+\frac{79z^9+776z^7+1482z^5-1920z^3-945z}{92160n^4}+$$
$$\frac{9z^{11}+113z^9+310z^7-594z^5-255z^3+5985z}{122880n^5}+\frac{1}{185794560n^6}(1065z^{13}+15448z^{11}+48821z^9-82440z^7+616707z^5+6667920z^3+2463615z)+$$
$$\frac{1}{743178240n^7}(339z^{15}+6891z^{13}+41107z^{11}+113891z^9+1086849z^7+5639193z^5-18226215z^3-111486375z)+$$
$$\frac{1}{356725555200n^8}(9159z^{17}+296624z^{15}+3393364z^{13}+16657824z^{11}+27817290z^9-591760080z^7-9178970220z^5-42618441600z^3-14223634425z)+$$
$$\frac{1}{1426902220800n^9}(63z^{19}-7857z^{17}-131468z^{15}-5104636z^{13}-115962198z^{11}-1311524070z^9-8066259180z^7-5512748220z^5+294835704975z^3+1221207562575z)$$

We get Cornish-Fisher to a high order with little effort! This last formula extends (and corrects) Equation (75) of Shaw (JCF, 2006), and extends the published results (see e.g. A&S). It is an example for the Student but we can do this trick for all kinds of pairs. But note also we did NOT have to expand in powers of n: we have a raw form as:

```
rawseries = Series[rawform, {z, 0, 19}];
```

```
Series[rawseries, {z, 0, 5}]
```

$$\frac{\sqrt{n}\,\Gamma\left(\frac{n}{2}\right) z}{\sqrt{2}\,\Gamma\left(\frac{n+1}{2}\right)} + \left(\frac{\sqrt{n}\,(n+1)\,\Gamma\left(\frac{n}{2}\right)^3}{12\sqrt{2}\,\Gamma\left(\frac{n+1}{2}\right)^3} - \frac{\sqrt{n}\,\Gamma\left(\frac{n}{2}\right)}{6\sqrt{2}\,\Gamma\left(\frac{n+1}{2}\right)}\right) z^3 + \left(\frac{\sqrt{n}\,(7n^2+8n+1)\,\Gamma\left(\frac{n}{2}\right)^5}{480\sqrt{2}\,\Gamma\left(\frac{n+1}{2}\right)^5} - \frac{\sqrt{n}\,(n+1)\,\Gamma\left(\frac{n}{2}\right)^3}{24\sqrt{2}\,\Gamma\left(\frac{n+1}{2}\right)^3} + \frac{\sqrt{n}\,\Gamma\left(\frac{n}{2}\right)}{40\sqrt{2}\,\Gamma\left(\frac{n+1}{2}\right)}\right) z^5 + O(z^6)$$

Much of the complication and lack of precision in the Cornish-Fisher expansion arises from it having an unnecessary expansion of all these gamma functions in inverse powers of n. We have literally added up this part of the expansion. Let's do it for a decidedly non-asymptotic $n = 3$! This is a good case to consider as the Quantile Functions are all known in closed form for $n = 1, 2, 4$ and the cases of even n are simple polynomial-solution problems. We look at the errors compared to the exact quantile function.

```
squeezer[z_] = N[Normal[rawseries /. n → 3], 20];
Squeezer[z_] := squeezer[z];
Squeezer[z]
```

$$1.0814573014721493537 \times 10^{-13} z^{19} + 2.3265242362709033219 \times 10^{-12} z^{17} + 3.0876508827972676651 \times 10^{-10} z^{15} + 1.3313226371711078036 \times 10^{-8} z^{13} + 4.4524211572805376698 \times 10^{-7} z^{11} + 0.000014790364130687491724 z^9 + 0.00040346678684943922555 z^7 + 0.0078035220067134673222 z^5 + 0.10325723454150953095 z^3 + 1.0854018818374014890 z$$

Note that there is NO approximation in the coefficients here. The only approximation is in the number of terms we take.

```
ExactQuantileF[y_, n_] := Module[{arg = If[y < 1 / 2, 2 y, 2 (1 - y)]},
  Sign[y - 1 / 2] Sqrt[n * (1 / InverseBetaRegularized[arg, n / 2, 1 / 2] - 1)]]
```

```
explot = Plot[{ExactQuantileF[u, 3], Squeezer[Sqrt[2] * InverseErf[2 * u - 1]]}, {u, 0.0001, 0.9999}, PlotRange → All];
```



There really are two plots here, but we cannot see the difference from this overlay. So we plot the difference, and we have to look well into the tail to see it:

```
explot = Plot[ExactQuantileF[u, 3] - Squeezer[Sqrt[2] * InverseErf[2 * u - 1]], {u, 0.99, 0.9999}, PlotRange → All];
```



These are absolute errors - the relative error is much smaller. So we have the possibility of using truncated polynomial expressions of Cornish-Fisher type in non-asymptotic domain. The errors are under investigation, as is the possibility of using a better sample transmutation mapping in the tail region.

We also have a variation of a multivariate T with general degrees of freedom in the marginals, based on a Gaussian copula - just squeeze the mixed Gaussians using a high-order Cornish-Fisher expansion.

```
explot = Plot[100 * (Squeezer[Sqrt[2] * InverseErf[2 * u - 1]] / ExactQuantileF[u, 3] - 1), {u, 0.9999, 0.9999999}, PlotRange → All];
```



Example 2: Rank Transmutation and introduction of skewness into any distribution

To define a *rank transmutation mapping* in complete generality, suppose that we have two distributions with a common sample space, with CDFs F_1 and F_2. We can form

$$G_{R12}(u) = F_2(F_1^{-1}(u)), \quad G_{R21}(u) = F_1(F_2^{-1}(u)) \tag{10}$$

and this pair of maps the unit interval $I = [0, 1]$ into itself and under suitable assumptions are mutual inverses and satisfy $G_{ij}(0) = 0$, $G_{ij}(1) = 1$. We shall further assume that these rank transmuation maps are continuously differentiable, as otherwise it is easy to check that a transmuted density may be discontinuous. We could consider all kinds of such maps arising from a particular choice of the F_i, but here, rather, we shall postulate an interesting form. Consider, for $|\lambda| \leq 1$,

$$G_{R12}(u) = u + \lambda u (1 - u) \tag{11}$$

This has the consequence that the CDFs are related by

$$F_2(x) = (1 + \lambda) F_1(x) - \lambda F_1(x)^2 \tag{12}$$

and *the sampling algorithm remains tractable* as the Quantile functions are related by

$$F_2^{-1}(u) = F_1^{-1}\,(G_{R21}(u)), \quad G_{R21}(u) \;=\; \frac{1+\lambda-\sqrt{(\lambda+1)^2-4\,\lambda\,u}}{2\,\lambda} \tag{13}$$

There are two important extremal cases. First, if $\lambda = -1$, then $G_{R12}(u) = u^2$ and $F_2(x) \;=\; F_1(x)^2$ and we recognize that the ditribution of F_2 corresponds to that of the maximum of two independent copies of the F_1distribution. Correspondingly $\lambda = +1$ generates the distribution of the minimum. We could use higher powers and distributions of max/min of k. But the quadratic is tractable and natural, and is easily seen to give us a natural skewing mechanism.

■ A Skew-Uniform example

```
H[z_] := z; h[z_] := 1;
H2[z_, λ_] := (1 + λ) H[z] - λ H[z]^2;
h2a[x_, λ_] = D[H2[x, λ], x];
h2[x_, λ_] := h2a[x, λ]
```

```
h2[x, p]
```

$$-2\,x\,p + p + 1$$

```
Do[Plot[h2[x, p], {x, 0, 1}, PlotRange → {0, 2}], {p, -1, 1, 0.5}]
```


2
1.75
1.5
1.25
1
0.75
0.5
0.25
0.2
0.4
0.6
0.8
1


2
1.75
1.5
1.25
1
0.75
0.5
0.25
0.2
0.4
0.6
0.8
1


2
1.75
1.5
1.25
1
0.75
0.5
0.25
0.2
0.4
0.6
0.8
1


2
1.75
1.5
1.25
1
0.75
0.5
0.25
0.2
0.4
0.6
0.8
1



This is a very natural interpretation of a "skew-uniform".

■ A Skew-Normal example

```
Φ[z_] := 1 / 2 (1 + Erf[z / Sqrt[2]]);
```

```
f2[x_, λ_] := 1 / Sqrt[2 Pi] Exp[-x^2 / 2] (1 + λ - 2 λ Φ[x])
```

The first 4 moments are then

```
fourmoments = Integrate[{x, x^2, x^3, x^4} f2[x, λ], {x, -Infinity, Infinity}]
```

$$\left\{-\frac{\lambda}{\sqrt{\pi}}, 1, -\frac{5\,\lambda}{2\,\sqrt{\pi}}, 3\right\}$$

so that the variance is

```
fourmoments[[2]] - fourmoments[[1]]^2
```

$$1 - \frac{\lambda^2}{\pi}$$

and the skewness is

```
Integrate[(x - fourmoments[[1]])^3 ftwo[x, λ], {x, -Infinity, Infinity}]
```

$$\frac{\lambda\,(\pi - 4\,\lambda^2)}{2\,\pi^{3/2}}$$

```
Plot[Evaluate[Table[ftwo[x, λ], {λ, 0, 1, 0.5}]], {x, -4, 4}];
```



If we want to standardize this distribution to have zero mean and unit variance we form the following from an obvious linear transformation:

```
f3[x_, λ_] := Sqrt[1 - λ^2 / Pi] * f2[x Sqrt[1 - λ^2 / Pi] - λ / Sqrt[Pi], λ]
```

We could, if we wanted to, do an series expansion of this around the standard Normal distribution, in powers of both x and λ, in order to see the relation to an Gram-Charlier-type expansion, but, as with the Cornish-Fisher expansion on the Quantile, this is now unnecessary!

We could also just as easily do skew-Student or skew anything else, and the Monte Carlo sampling is precisely as tractable as it is for the base distribution. These ideas should of course be compared with the work of many others, in particular A. Azzalini and co-workers, who use a different mechanism. See his web site at http://tango.stat.unipd.it/SN/ for details of their approach, which is based on multiplying a density by a CDF. Our approach coincides with his in certain special cases (e.g. distributions of max/min are also contained) but appears to be different in general.

Summary

Computer algebra is a powerful technique for:

(a) generating quantile functions in the first place.

(b) turning samples from one distribution into another;

(c) turning the ranks of one distribution into the ranks of another, e,g, to introduce skewness in a universal way;

These techniques are well adapted for Quasi-Monte-Carlo and Copula simulation methods.